Computergestuetzte Mathematik zur Analysis

Lektion 14 (30. Januar)

```
> restart:
```

Gewöhnliche Differentialgleichungen II

```
> os := diff(y(x),x$2) + y(x);
```

$$os := \frac{d^2}{dx^2}\, y(x) + y(x) \tag{1.1}$$

```
> dsolve(os=0,y(x));
```

$$y(x) = \_C1 \sin(x) + \_C2 \cos(x) \tag{1.2}$$

```
> dsolve({os=0,y(0)=1,D(y)(0)=0},y(x));
```

$$y(x) = \cos(x) \tag{1.3}$$

```
> l1 := rhs((1.3));
```

$$l1 := \cos(x) \tag{1.4}$$

```
> gos:= diff(y(x),x$2) + 1/5*diff(y(x),x) + y(x);
```

$$gos := \frac{d^2}{dx^2}\, y(x) + \frac{1}{5} \frac{d}{dx}\, y(x) + y(x) \tag{1.5}$$

```
> dsolve({gos=0,y(0)=1,D(y)(0)=0},y(x));
```

$$y(x) = \frac{1}{33} \sqrt{11}\, e^{-\frac{1}{10} x} \sin\left(\frac{3}{10} \sqrt{11}\, x\right) + e^{-\frac{1}{10} x} \cos\left(\frac{3}{10} \sqrt{11}\, x\right) \tag{1.6}$$

```
> l2 := rhs((1.6));
```

$$l2 := \frac{1}{33} \sqrt{11}\, e^{-\frac{1}{10} x} \sin\left(\frac{3}{10} \sqrt{11}\, x\right) + e^{-\frac{1}{10} x} \cos\left(\frac{3}{10} \sqrt{11}\, x\right) \tag{1.7}$$

```
> plot([l1,l2],x=0..20);
```



Inhomogene Gewöhnliche Differentialgleichungen

```
> l3:=rhs(dsolve({os=cos(1*x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l3 := \cos(x) + \frac{1}{2}\sin(x)\,x \qquad \textbf{(2.1)}$$

```
> plot(l3,x=0..100); # Resonanzfall
```




```
> l4:= rhs(dsolve({os=sin(3/4*x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l4 := -\frac{12}{7}\sin(x)+\cos(x)+\frac{16}{7}\sin\left(\frac{3}{4}x\right) \quad \textbf{(2.2)}$$

```
> l5:= rhs(dsolve({os=sin(7/8*x),y(0)=1,D(y)(0)=0},y(x)));
```

$$l5 := -\frac{56}{15}\sin(x)+\cos(x)+\frac{64}{15}\sin\left(\frac{7}{8}x\right) \quad \textbf{(2.3)}$$

```
> plot([l4,l5],x=0..100,color=[red,blue]);
```



Bessel Funktionen

```
> with(VectorCalculus):
> SetCoordinates(polar);
```

$$\text{polar} \tag{3.1}$$

```
> Laplacian(u(r,phi),[r,phi]);
```

$$\frac{\frac{\partial}{\partial r} u(r, \phi) + r \left(\frac{\partial^2}{\partial r^2} u(r, \phi) \right) + \frac{\frac{\partial^2}{\partial \phi^2} u(r, \phi)}{r}}{r} \tag{3.2}$$

```
> LG:=Laplacian(v(r)*w(phi),[r,phi])+v(r)*w(phi);
```

$$LG := \frac{\left(\frac{d}{dr} v(r) \right) w(\phi) + r \left(\frac{d^2}{dr^2} v(r) \right) w(\phi) + \frac{v(r) \left(\frac{d^2}{d\phi^2} w(\phi) \right)}{r}}{r} + v(r)\, w(\phi) \tag{3.3}$$

```
> isolate(expand(LG*r^2/v(r)/w(phi)),r);
```

$$\frac{r\left(v(r)\,r+\left(\frac{d^2}{dr^2}\,v(r)\right)r+\frac{d}{dr}\,v(r)\right)}{v(r)}=-\frac{\frac{d^2}{d\phi^2}\,w(\phi)}{w(\phi)} \tag{3.4}$$

```
> collect(lhs((3.4))*v(r)-n^2*v(r),v(r));
```

$$\left(-n^2+r^2\right)v(r)+r\left(\left(\frac{d^2}{dr^2}\,v(r)\right)r+\frac{d}{dr}\,v(r)\right) \tag{3.5}$$

```
> g:= x^2 * diff(y(x),x$2)+x*diff(y(x),x) + (x^2-n^2)*y(x);
```

$$g:=x^2\left(\frac{d^2}{dx^2}\,y(x)\right)+x\left(\frac{d}{dx}\,y(x)\right)+\left(-n^2+x^2\right)y(x) \tag{3.6}$$

```
> dsolve(g=0,y(x));
```

$$y(x)=\_C1\,\mathrm{BesselJ}(n,x)+\_C2\,\mathrm{BesselY}(n,x) \tag{3.7}$$

```
> farben:=[red,blue,cyan,magenta];
```

$$farben:=[red,\ blue,\ cyan,\ magenta] \tag{3.8}$$

```
> plot([seq(BesselJ(n,x),n=0..3)],x=0..15,color=farben);
```




```
> plot([seq(BesselY(n,x),n=0..3)],x=0..15,y=-1..0.6,color=farben)
;
```




```
> ns1:= seq(fsolve(BesselJ(1,x)=0,x,3.5+3*(k-1)..3.5+3*k),k=1..4)
  ;
```

$$ns1 := 3.831705970, 7.015586670, 10.17346814, 13.32369194 \quad \textbf{(3.9)}$$

```
> fnm1 := [ r*cos(s), r*sin(s), BesselJ(1,ns1[2]*r)*cos(s)];
```

$$fnm1 := [r\cos(s), r\sin(s), \mathrm{BesselJ}(1, 7.015586670\, r)\cos(s)] \quad \textbf{(3.10)}$$

```
> plotarg := style=patchcontour,orientation=[120,60];
```

$$plotarg := style = patchcontour, orientation = [120, 60] \quad \textbf{(3.11)}$$

```
> plot3d(fnm1,r=0..1,s=0..2*Pi,plotarg);
```




```
> ns2 := fsolve(BesselJ(2,x)=0,x,8..9);
```

$$ns2 := 8.417244140 \quad (3.12)$$

```
> fnm:= [ r*cos(s), r*sin(s), BesselJ(2,ns2*r)*cos(2*s)];
```

$$fnm := [r\cos(s), r\sin(s), \text{BesselJ}(2, 8.417244140\, r)\cos(2\, s)] \quad (3.13)$$

```
> plot3d(fnm,r=0..1,s=0..2*Pi,plotarg);
```




```
> fnm:= [ r*cos(s), r*sin(s), sin(2*k*Pi/21)*BesselJ(2,ns2*r)*cos
  (2*s)];
```

$$fnm := \left[r\cos(s),\ r\sin(s),\ \sin\left(\frac{2}{21}\,k\pi\right)\mathrm{BesselJ}(2,\ 8.417244140\ r)\cos(2\ s) \right] \tag{3.14}$$

```
> with(plots):
> animate(plot3d,[fnm,r=0..1,s=0..2*Pi,plotarg],k=0..20);
```

$k = 0.$



Differentialgleichungssysteme

```
> restart;
> with(LinearAlgebra):
> A:=<<0|1|0>,<-1|0|1>,<0|0|2>>;
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 1 \\ 0 & 0 & 2 \end{bmatrix} \quad (4.1)$$

```
> T:=MatrixExponential(A,t);
```

$$T := \begin{bmatrix} \cos(t) & \sin(t) & -\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}e^{2t} \\ -\sin(t) & \cos(t) & -\frac{2}{5}\cos(t) + \frac{2}{5}e^{2t} + \frac{1}{5}\sin(t) \\ 0 & 0 & e^{2t} \end{bmatrix} \quad (4.2)$$

```
> #Loesung y' = A*y , y(0) = <a,b,c>
```

```
> y0 := <a,b,c>;
```

$$y0 := \begin{bmatrix} a \\ b \\ c \end{bmatrix} \quad \textbf{(4.3)}$$

```
> y(t) := T.y0;
```

$$y(t) := \begin{bmatrix} \cos(t)\, a + \sin(t)\, b + \left(-\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}\,e^{2t}\right) c \\ -\sin(t)\, a + \cos(t)\, b + \left(-\frac{2}{5}\cos(t) + \frac{2}{5}\,e^{2t} + \frac{1}{5}\sin(t)\right) c \\ e^{2t}\, c \end{bmatrix} \quad \textbf{(4.4)}$$

```
> with(VectorCalculus):
> BasisFormat(false):
> diff(y(t),t) - A.y(t);
```

$$\Big[\big[0\big], \quad \textbf{(4.5)}$$

$$\left[\left(\frac{2}{5}\sin(t) + \frac{4}{5}\,e^{2t} + \frac{1}{5}\cos(t)\right) c + \left(-\frac{2}{5}\sin(t) - \frac{1}{5}\cos(t) + \frac{1}{5}\,e^{2t}\right) c \right.$$

$$\left. - e^{2t}\, c\right],$$

$$\big[0\big]\Big]$$

```
> simplify((4.5));
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(4.6)}$$

```
> eval(y(t),t=0);
```

$$\begin{bmatrix} a \\ b \\ c \end{bmatrix} \quad \textbf{(4.7)}$$

Das Pendel

```
> restart;
> Dgl := diff(y(t),t$2)= -sin(y(t));
```

$$Dgl := \frac{d^2}{dt^2}\, y(t) = -\sin(y(t)) \quad \textbf{(5.1)}$$

```
> AW:= y(0)=Pi/8,D(y)(0)=0;
```

$$AW := y(0) = \frac{1}{8}\pi,\ D(y)(0) = 0 \quad \textbf{(5.2)}$$

```
> dsolve({Dgl,AW},y(t));
```

$$y(t) = RootOf\left(\int_{\_Z}^{\frac{1}{8}\pi} \frac{1}{\sqrt{2\cos(\_a) - 2\cos\left(\frac{1}{8}\pi\right)}}\, d\_a + t\right), y(t) = RootOf\left(\int_{\frac{1}{8}\pi}^{\_Z} \frac{1}{\sqrt{2\cos(\_a) - 2\cos\left(\frac{1}{8}\pi\right)}}\, d\_a + t\right) \quad (5.3)$$

```
> Lsg:=dsolve({Dgl,AW},y(t),type=numeric,output=listprocedure);
```

$Lsg := \left[t = \textbf{proc}(t)\ \ldots\ \textbf{end proc}, y(t) = \textbf{proc}(t)\ \ldots\ \textbf{end proc}, \frac{d}{dt}\, y(t) = \right.$ **(5.4)**

proc(t)

...

end proc]

```
> yl := eval(y(t),Lsg);
```

$$yl := \textbf{proc}(t)\ \ldots\ \textbf{end proc} \quad (5.5)$$

```
> yl(1);
```

$$0.215837134280979 \quad (5.6)$$

```
> Dgl_os := diff(y(t),t$2) = -y(t);
```

$$Dgl\_os := \frac{d^2}{dt^2}\, y(t) = -y(t) \quad (5.7)$$

```
> dsolve({Dgl_os,AW},y(t));
```

$$y(t) = \frac{1}{8}\pi\cos(t) \quad (5.8)$$

```
> yl_os:=unapply(rhs((5.8)),t);
```

$$yl\_os := t \rightarrow \frac{1}{8}\pi\cos(t) \quad (5.9)$$

```
> plot([yl,yl_os],0..50,color=[black,red]);
```




```
> AW2:= y(0)=Pi/4,D(y)(0)=0;
```

$$AW2 := y(0) = \frac{1}{4}\pi, D(y)(0) = 0 \quad \textbf{(5.10)}$$

```
> Lsg:=dsolve({Dgl,AW2},y(t),type=numeric,output=listprocedure);
```

$$Lsg := \left[t = \textbf{proc}(t) \ \ldots \ \textbf{end proc}, y(t) = \textbf{proc}(t) \ \ldots \ \textbf{end proc}, \frac{d}{dt} y(t) = \right. \quad \textbf{(5.11)}$$

proc(t)

...

end proc]

```
> yl := eval(y(t),Lsg);
```

$$yl := \textbf{proc}(t) \ \ldots \ \textbf{end proc} \quad \textbf{(5.12)}$$

```
> dsolve({Dgl_os,AW2},y(t));
```

$$y(t) = \frac{1}{4}\pi\cos(t) \quad \textbf{(5.13)}$$

```
> yl_os:=unapply(rhs((5.13)),t);
```

$$yl\_os := t \rightarrow \frac{1}{4}\pi\cos(t) \quad \textbf{(5.14)}$$

```
> plot([yl,yl_os],0..50,color=[black,red]);
```